
F08 – Least Squares and Eigenvalue Problems (LAPACK)

F08NVF (CGEBAL/ZGEBAL) – NAG Fortran Library Routine Document

Note. Before using this routine, please read the Users’ Note for your implementation to check the interpretation of bold
italicised terms and other implementation-dependent details.

1 Purpose

F08NVF (CGEBAL/ZGEBAL) balances a complex general matrix in order to improve the accuracy of
computed eigenvalues and/or eigenvectors.

2 Specification

SUBROUTINE F08NVF(JOB, N, A, LDA, ILO, IHI, SCALE, INFO)
ENTRY cgebal(JOB, N, A, LDA, ILO, IHI, SCALE, INFO)
INTEGER N, LDA, ILO, IHI, INFO
real SCALE(∗)
complex A(LDA,∗)
CHARACTER∗1 JOB

The ENTRY statement enables the routine to be called by its LAPACK name.

3 Description

This routine balances a complex general matrix A. The term ‘balancing’ covers two steps, each of which
involves a similarity transformation of A. The routine can perform either or both of these steps.

1. The routine first attempts to permute A to block upper triangular form by a similarity
transformation:

PAPT = A′ =




A′
11 A′

12 A′
13

0 A′
22 A′

23

0 0 A′
33




where P is a permutation matrix, and A′
11 and A′

33 are upper triangular. Then the diagonal elements
of A′

11 and A′
33 are eigenvalues of A. The rest of the eigenvalues of A are the eigenvalues of the

central diagonal block A′
22, in rows and columns ilo to ihi. Subsequent operations to compute the

eigenvalues of A (or its Schur factorization) need only be applied to these rows and columns; this
can save a significant amount of work if ilo > 1 and ihi < n. If no suitable permutation exists (as is
often the case), the routine sets ilo = 1 and ihi = n, and A′

22 is the whole of A.

2. The routine applies a diagonal similarity transformation to A′, to make the rows and columns of
A′

22 as close in norm as possible:

A′′ = DA′D−1 =




I 0 0
0 D22 0
0 0 I

33







I 0 0
0 D−1

22 0
0 0 I


 .

This scaling can reduce the norm of the matrix (that is, ‖ A′′
22 ‖ < ‖ A′

22 ‖), and hence reduce the
effect of rounding errors on the accuracy of computed eigenvalues and eigenvectors.

4 References

[1] Golub G H and van Loan C F (1996) Matrix Computations Johns Hopkins University Press (3rd
Edition), Baltimore

5 Parameters

1: JOB — CHARACTER*1 Input

On entry: indicates whether A is to be permuted to and/or scaled (or neither), as follows:

if JOB = ’N’, then A is neither permuted nor scaled (but values are assigned to ILO, IHI and
SCALE);
if JOB = ’P’, then A is permuted but not scaled;
if JOB = ’S’, then A is scaled but not permuted;
if JOB = ’B’, then A is both permuted and scaled.

Constraint: JOB = ’N’, ’P’, ’S’ or ’B’.

2: N — INTEGER Input

On entry: n, the order of the matrix A.

Constraint: N ≥ 0.

3: A(LDA,∗) — complex array Input/Output

Note: the second dimension of the array A must be at least max(1,N).

On entry: the n by n general matrix A.

On exit: A is overwritten by the balanced matrix.

A is not referenced if JOB = ’N’.

4: LDA — INTEGER Input

On entry: the first dimension of the array A as declared in the (sub)program from which F08NVF
(CGEBAL/ZGEBAL) is called.

Constraint: LDA ≥ max(1,N).

5: ILO — INTEGER Output
6: IHI — INTEGER Output

On exit: the values ilo and ihi such that on exit A(i, j) is zero if i > j and 1 ≤ j < ilo or ihi < i ≤ n.

If JOB = ’N’ or ’S’, then ilo = 1 and ihi = n.

7: SCALE(∗) — real array Output

Note: the dimension of the array SCALE must be at least max(1,N).

On exit: details of the permutations and scaling factors applied to A. More precisely, if pj is the
index of the row and column interchanged with row and column j and dj is the scaling factor used
to balance row and column j, then

SCALE(j) =




pj , forj = 1, 2, . . . , ilo − 1
dj , forj = ilo, ilo + 1, . . . , ihi and
pj , forj = ihi + 1, ihi + 2, . . . , n.

The order in which the interchanges are made is n to ihi + 1, then 1 to ilo − 1.

8: INFO — INTEGER Output

On exit: INFO = 0 unless the routine detects an error (see Section 6).

6 Error Indicators and Warnings

INFO < 0

If INFO = −i, the ith parameter had an illegal value. An explanatory message is output, and
execution of the program is terminated.

7 Accuracy

The errors are negligible, compared with those in subsequent computations.

8 Further Comments

If the matrix A is balanced by this routine, then any eigenvectors computed subsequently are eigenvectors
of the matrix A′′ (see Section 3) and hence F08NWF (CGEBAK/ZGEBAK) must then be called to
transform them back to eigenvectors of A.

If the Schur vectors of A are required, then this routine must not be called with JOB = ’S’ or ’B’, because
then the balancing transformation is not unitary. If this routine is called with JOB = ’P’, then any Schur
vectors computed subsequently are Schur vectors of the matrix A′′, and F08NWF must be called (with
SIDE = ’R’) to transform them back to Schur vectors of A.

The total number of real floating-point operations is approximately proportional to n2.

The real analogue of this routine is F08NHF (SGEBAL/DGEBAL).

9 Example

To compute all the eigenvalues and right eigenvectors of the matrix A, where

A =




1.50− 2.75i 0.00 + 0.00i 0.00 + 0.00i 0.00 + 0.00i
−8.06− 1.24i −2.50− 0.50i 0.00 + 0.00i −0.75 + 0.50i
−2.09 + 7.56i 1.39 + 3.97i −1.25 + 0.75i −4.82− 5.67i
6.18 + 9.79i −0.92− 0.62i 0.00 + 0.00i −2.50− 0.50i


 .

The program first calls F08NVF (CGEBAL/ZGEBAL) to balance the matrix; it then computes the
Schur factorization of the balanced matrix, by reduction to Hessenberg form and the QR algorithm.
Then it calls F08QXF (CTREVC/ZTREVC) to compute the right eigenvectors of the balanced matrix,
and finally calls F08NWF (CGEBAK/ZGEBAK) to transform the eigenvectors back to eigenvectors of
the original matrix A.

9.1 Program Text

Note. The listing of the example program presented below uses bold italicised terms to denote precision-dependent details.
Please read the Users’ Note for your implementation to check the interpretation of these terms. As explained in the Essential
Introduction to this manual, the results produced may not be identical for all implementations.

* F08NVF Example Program Text
* Mark 16 Release. NAG Copyright 1992.
* .. Parameters ..

INTEGER NIN, NOUT
PARAMETER (NIN=5,NOUT=6)
INTEGER NMAX, LDA, LDH, LWORK, LDVL, LDVR
PARAMETER (NMAX=8,LDA=NMAX,LDH=NMAX,LWORK=64*NMAX,LDVL=1,

+ LDVR=NMAX)
* .. Local Scalars ..

INTEGER I, IFAIL, IHI, ILO, INFO, J, M, N
* .. Local Arrays ..

complex A(LDA,NMAX), H(LDH,NMAX), TAU(NMAX), VL(LDVL,1),
+ VR(LDVR,NMAX), W(NMAX), WORK(LWORK)

real RWORK(NMAX), SCALE(NMAX)
LOGICAL SELECT(1)
CHARACTER CLABS(1), RLABS(1)

* .. External Subroutines ..
EXTERNAL F06TFF, X04DBF, cgebak, cgebal, cgehrd, chseqr,

+ ctrevc, cunghr
* .. Intrinsic Functions ..

INTRINSIC real, imag
* .. Executable Statements ..

WRITE (NOUT,*) ’F08NVF Example Program Results’
* Skip heading in data file

READ (NIN,*)
READ (NIN,*) N
IF (N.LE.NMAX) THEN

*
* Read A from data file
*

READ (NIN,*) ((A(I,J),J=1,N),I=1,N)
*
* Balance A
*

CALL cgebal(’Both’,N,A,LDA,ILO,IHI,SCALE,INFO)
*
* Reduce A to upper Hessenberg form H = (Q**H)*A*Q
*

CALL cgehrd(N,ILO,IHI,A,LDA,TAU,WORK,LWORK,INFO)
*
* Copy A to H
*

CALL F06TFF(’General’,N,N,A,LDA,H,LDH)
*
* Copy A into VR
*

CALL F06TFF(’General’,N,N,A,LDA,VR,LDVR)
*
* Form Q explicitly, storing the result in VR
*

CALL cunghr(N,1,N,VR,LDVR,TAU,WORK,LWORK,INFO)
*
* Calculate the eigenvalues and Schur factorization of A
*

CALL chseqr(’Schur form’,’Vectors’,N,ILO,IHI,H,LDH,W,VR,LDVR,
+ WORK,LWORK,INFO)

*
WRITE (NOUT,*)
IF (INFO.GT.0) THEN

WRITE (NOUT,*) ’Failure to converge.’
ELSE

WRITE (NOUT,*) ’Eigenvalues’
WRITE (NOUT,99999) (’ (’,real(W(I)),’,’,imag(W(I)),’)’,I=1,

+ N)
*
* Calculate the eigenvectors of A, storing the result in VR
*

CALL ctrevc(’Right’,’Overwrite’,SELECT,N,H,LDH,VL,LDVL,VR,
+ LDVR,N,M,WORK,RWORK,INFO)

*
* Backtransform eigenvectors

*
CALL cgebak(’Both’,’Right’,N,ILO,IHI,SCALE,M,VR,LDVR,INFO)

*
* Print eigenvectors
*

WRITE (NOUT,*)
IFAIL = 0

*
CALL X04DBF(’General’,’ ’,N,M,VR,LDVR,’Bracketed’,’F7.4’,

+ ’Contents of array VR’,’Integer’,RLABS,
+ ’Integer’,CLABS,80,0,IFAIL)

*
END IF

END IF
STOP

*
99999 FORMAT ((3X,4(A,F7.4,A,F7.4,A,:)))

END

9.2 Program Data

F08NVF Example Program Data
4 :Value of N

( 1.50,-2.75) ( 0.00, 0.00) ( 0.00, 0.00) ( 0.00, 0.00)
(-8.06,-1.24) (-2.50,-0.50) ( 0.00, 0.00) (-0.75, 0.50)
(-2.09, 7.56) ( 1.39, 3.97) (-1.25, 0.75) (-4.82,-5.67)
( 6.18, 9.79) (-0.92,-0.62) ( 0.00, 0.00) (-2.50,-0.50) :End of matrix A

9.3 Program Results

F08NVF Example Program Results

Eigenvalues
(-1.2500, 0.7500) (-1.5000,-0.4975) (-3.5000,-0.5025) ( 1.5000,-2.7500)

Contents of array VR
1 2 3 4

1 ( 0.0000, 0.0000) ( 0.0000, 0.0000) ( 0.0000, 0.0000) ( 0.1452, 0.0000)
2 ( 0.0000, 0.0000) (-0.0616, 0.0413) ( 0.4613, 0.0000) (-0.2072,-0.2450)
3 ( 1.0000, 0.0000) ( 0.6032,-0.3968) ( 0.2983, 0.7017) ( 0.7768, 0.2232)
4 ( 0.0000, 0.0000) ( 0.0822, 0.0000) ( 0.4251, 0.2850) (-0.0119, 0.4372)

